Одномерные и двумерные массивы таблицы

ОДНОМЕРНЫЕ И ДВУМЕРНЫЕ МАССИВЫ (ТАБЛИЦЫ)

Массив — это пронумерованная последовательность величин одинакового типа, обозначаемая одним именем. Элементы массива располагаются в последовательных ячейках памяти, обозначаются именем массива и индексом. Каждое из значений, составляющих массив, называется его компонентой (или элементом массива).

Массив данных в программе рассматривается как переменная структурированного типа. Массиву присваивается имя, посредством которого можно ссылаться как на массив данных в целом, так и на любую из его компонент.

Переменные, представляющие компоненты массивов, называются переменными с индексами в отличие от простых переменных, представляющих в программе элементарные данные. Индекс в обозначении компонент массивов может быть константой, переменной или выражением порядкового типа.

Если за каждым элементом массива закреплен только один его порядковый номер, то такой массив называется линейным. Вообще количество индексов элементов массива определяет размерность массива. По этом признаку массивы делятся на одномерные (линейные), двумерные, трёхмерные и т.д.

Пример: числовая последовательность четных натуральных чисел 2, 4, 6, ..., N представляет собой линейный массив, элементы которого можно обозначить А[1]=2, А[2]=4, А[3]=6, ..., А[К]=2*(К+1), где К — номер элемента, а 2, 4, 6, ..., N — значения. Индекс (порядковый номер элемента) записывается в квадратных скобках после имени массива.

Например, A[7] — седьмой элемент массива А; D[6] — шестой элемент массива D.

Для размещения массива в памяти ЭВМ отводится поле памяти, размер которого определяется типом, длиной и количеством компонент массива. В языке Pascal эта информация задается в разделе описаний. Массив описывается так:

	имя массива : Array [начальное значение индекса..конечное значение индекса] Of базовый тип;

Например, 

	Var B : Array [1..5] Of Real, R : Array [1..34] Of Char;

— описывается массив В, состоящий из 5 элементов и символьный массив R, состоящий из 34 элементов. Для массива В будет выделено 5*6=30 байт памяти, для массива R — 1*34=34 байта памяти. 

Базовый тип элементов массива может быть любым, за исключением файлового.

Заполнить массив можно следующим образом:

1) с помощью оператора присваивания. Этот способ заполнения элементов массива особенно удобен, когда между элементами существует какая-либо зависимость, например, арифметическая или геометрическая прогрессии, или элементы связаны между собой реккурентным соотношением.

Задача 1. Заполнить одномерный массив элементами, отвечающими следующему соотношению:

a1=1; a2=1; ai=ai-2+ai-1 (i = 3, 4, ..., n). 

Read(N); {Ввод количества элементов}

A[1]:= 1;

A[2]:= 1;

FOR I := 3 TO N DO

     A[I] := A[I - 1] + A[I - 2];

Другой вариант присваисвания значений элементам массива — заполнение значениями, полученными с помощью датчика случайных чисел.

Задача 2. Заполнить одномерный массив с помощью датчика случайных чисел таким образом, чтобы все его элементы были различны.

Program Create;

Type Mas = Array[1..100] Of Integer;

Var A : Mas; I, J, N : Byte; Log : Boolean;

Begin

    Write(''); ReadLn(N);

    randomize; A[1] := -32768 + random(65535);

    For I := 2 To N Do

    Begin

        Log := True;

        Repeat

            A[i] := -32768 + random(65535); J := 1;

            While Log and (j <= i - 1) Do

            begin Log := a[i] <> a[j]; j := j + 1 End

        Until Log

    End;

    For i := 1 to N Do Write(a[i]:7); writeln

End.

2) ввод значений элементов массива с клавиатуры используется обычно тогда, когда между элементами не наблюдается никакой зависимости. Например, последовательность чисел 1, 2, -5, 6, -111, 0 может быть введена в память следующим образом:

Program Vvod;

 Var N, I : Integer;

     A : Array [1..20] Of Integer;

Begin

  Write('Введите количество элементов массива ');  ReadLn(N);

  FOR I := 1 TO N DO

  Begin

   Write('Введите A[', I, '] '); ReadLn(A[I])

End.

Над элементами массивами чаще всего выполняются такие действия, как

а) поиск значений;

б) сортировка элементов в порядке возрастания или убывания;

в) подсчет элементов в массиве, удовлетворяющих заданному условию.

Cумму элементов массива можно подсчитать по формуле S=S+A[I] первоначально задав S=0. Количество элементов массива можно подсчитать по формуле К=К+1, первоначально задав К=0. Произведение элементов массива можно подсчитать по формуле P = P * A[I], первоначально задав P = 1.

Задача 3. Дан линейный массив целых чисел. Подсчитать, сколько в нем различных чисел.

{Подсчет количества различных чисел в линейном массиве.

 ИДЕЯ РЕШЕНИЯ: заводим вспомогательный массив, элементами

 которого являются логические величины (False - если элемент

 уже встречался ранее, True - иначе)}

Program Razlichnye_Elementy;

Var I, N, K, Kol : Integer;

     A : Array [1..50] Of Integer;

     Lo : Array [1..50] Of Boolean;

Begin

  Write('Введите количество элементов массива: '); ReadLn(N);

    FOR I := 1 TO N DO

     Begin

      Write('A[', I, ']='); ReadLn (A[I]);

      Lo[I] := True; {Заполняем вспомогательный массив значениями True}

     End;

  Kol := 0; {переменная, в которой будет храниться количество различных чисел}

  FOR I := 1 TO N DO

   IF Lo[I] THEN

    Begin

     Kol := Kol + 1;

     FOR K := I TO N DO

 {Во вспомогательный массив заносим значение False,

  если число уже встречалось ранее или совпадает с текущим элементом A[I]}

       Lo[K] := (A[K] <> A[I]) And Lo[K];

    End;

  WriteLn('Количество различных чисел: ', Kol)

END.

Тест: N = 10; элементы массива - 1, 2, 2, 2, -1, 1, 0, 34, 3, 3. Ответ: 6.

Задача 4. Дан линейный массив. Упорядочить его элементы в порядке возрастания.

{Сортировка массива выбором (в порядке возрастания).

 Идея решения: пусть часть массива (по K-й элемент включительно)

 отсортирована. Нужно найти в неотсортированной части массива

 минимальный элемент и поменять местами с (K+1)-м}

Program Sortirovka;

Var N, I, J, K, Pr : Integer; A : Array [1..30] Of Integer;

Begin

Write('Введите количество элементов: '); ReadLn(N);

   For I := 1 To N Do

    Begin

     Write('Введите A[', I, '] '); Readln(A[I]);

    End;

   WriteLn;

   For I := 1 To N - 1 Do

    Begin

     K := I;

     For J := I + 1 To N Do If A[J] <= A[K] Then K := J;

     Pr := A[I]; A[I] := A[K]; A[K] := Pr;

    End;

    For I := 1 To N Do Write(A[I], ' ');

  End.

Тест: N = 10; элементы массива - 1, 2, 2, 2, -1, 1, 0, 34, 3, 3.

Ответ: -1, -1, 0, 1, 2, 2, 2, 3, 3, 34.

Если два массива являются массивами эквивалентых типов, то возможно присваивание одного массива другому. При этом все компоненты присваиваемого массива копируются в тот массив,оторому присваивается значение. Типы массивов будут эквивалентными, если эти массивы описываются совместно или описываются идентификатором одного и того же типа. Например, в описании 

Type Massiv = Array[1..10] Of Real;

Var A, B : Massiv; C, D : Array[1..10] Of Real; E : Array[1..10] Of Real;

типы переменных A, B эквивалентны, и поэтому данные переменные совместимы по присваиванию; тип переменных C, D также один и тот же, и поэтому данные переменные также совместны по присваиванию. Но тип переменных C, D не эквивалентен типам переменных A, B, E, поэтому, например, A и D не совместны по присваиванию. Эти особенности необходимо учитывать при работе с массивами. 

При решении практических задач часто приходится иметь дело с различными таблицами данных, математическим эквивалентом которых служат матрицы. Такой способ организации данных, при котором каждый элемент определяется номером строки и номером столбца, на пересечении которых он расположен, называется двумерным массивом или таблицей.

Например, данные о планетах Солнечной системы представлены следующей таблицей:

Планета

Расст. до Солнца

Относ. обьем

Относ. масса

Меркурий

57.9

0.06

0.05

Венера

108.2

0.92

0.81

Земля

149.6

1.00

1.00

Марс

227.9

0.15

0.11

Юпитер

978.3

1345.00

318.40

Сатурн

1429.3

767.00

95.20

Их можно занести в память компьютера, используя понятие двумерного массива. Положение элемента в массиве определяется двумя индексами. Они показывают номер строки и номер столбца. Индексы разделяются запятой. Например: A[7, 6], D[56, 47].

Заполняется двумерный массив аналогично одномерному: с клавиатуры, с помощью оператора присваивания. Например, в результате выполнения программы: 

Program Vvod2;

 Var I, J : Integer;

     A : Array [1..20, 1..20] Of Integer;

 Begin

  FOR I := 1 TO 3 DO

   FOR J := 1 TO 2 DO A[I, J] := 456 + I

 End.

элементы массива примут значения A[1, 1] = 457; A[1, 2] = 457; A[2, 1] = 458; A[2, 2] = 458; A[3, 1] = 459; A[3, 2] = 459. 

При описании массива задается требуемый объем памяти под двумерный массив, указываются имя массива и в квадратных скобках диапазоны изменения индексов.

При выполнении инженерных и математических расчетов часто используются переменные более чем с двумя индексами. При решении задач на ЭВМ такие переменные представляются как компоненты соответственно трех-, четырехмерных массивов и т.д.

Однако описание массива в виде многомерной структуры делается лишь из соображений удобства программирования как результат стремления наиболее точно воспроизвести в программе объективно существующие связи между элементами данных решаемой задачи. Что же касается образа массива в памяти ЭВМ, то как одномерные, так и многомерные массивы хранятся в виде линейной последовательности своих компонент, и принципиальной разницы между одномерными и многомерными массивами в памяти ЭВМ нет. Однако порядок, в котором запоминаются элементы многомерных массивов, важно себе представлять. В большинстве алгоритмических языков реализуется общее правило, устанавливающее порядок хранения в памяти элементов массивов: элементы многомерных массивов хранятся в памяти в последовательности, соответствующей более частому изменению младших индексов.

Задача 5. Заполнить матрицу порядка n по следующему образцу:

1

2

3

...

n-2

n-1

n

2

1

2

...

n-3

n-2

n-1

3

2

1

...

n-4

n-3

n-2

...

...

...

...

...

...

...

n-1

n-2

n-3

...

2

1

2

n

n-1

n-2

...

3

2

1

Program Massiv12;

  Var I, J, K, N : Integer; A : Array [1..10, 1..10] Of Integer;

  Begin

   Write('Введите порядок матрицы: '); ReadLn(N);

    For I := 1 To N Do

     For J := I To N Do

      Begin

       A[I, J] := J - I + 1; A[J, I] := A[I, J];

      End;

     For I := 1 To N Do

      Begin

       WriteLn;

       For J := 1 To N Do Write(A[I, J]:4);

      End

  End.

Задача 6. Дана целочисленная квадратная матрица. Найти в каждой строке наибольший элемент и поменять его местами с элементом главной диагонали.

Program Obmen;

 Var N, I, J, Max,Ind, Vsp : Integer;A : Array [1..15, 1..15] Of Integer;

 Begin

  WRITE('Введите количество элементов в массиве: '); READLN(N);

 FOR I := 1 TO N DO

  FOR J := 1 TO N DO

  Begin

   WRITE('A[', I, ',', J, '] '); READLN(A[I, J])

  End;

 FOR I := 1 TO N DO

 Begin

  Max := A[I, 1]; Ind := 1;

   FOR J := 2 TO N DO

      IF A[I, J] > Max THEN

       Begin

        Max := A[I, J]; Ind := J

       End;

   Vsp := A[I, I]; A[I, I] := A[I, Ind]; A[I, Ind] := Vsp

  End;

     FOR I := 1 TO N DO

      Begin

       WriteLn;

       FOR J := 1 TO N Do Write(A[I, J] : 3);

      End; WriteLn

 End.